Supporting Information

Recovery of Bulk Proton Magnetization and Sensitivity Enhancement in Ultrafast Magic-Angle Spinning Solid-State NMR

Jean-Philippe Demers†, Vinesh Vijayan‡, Adam Lange†*

Department of NMR-based Structural Biology, Max Planck Institute for Biophysical Chemistry, 37077 Göttingen, Germany

* Corresponding Author (demers@fmp-berlin.de)

† Department of Molecular Biophysics, Leibniz-Institut für Molekulare Pharmakologie (FMP), 13125 Berlin, Germany

‡ Department of Chemistry, IISER Thiruvananthapuram, 695016, Kerala, India

Bulk ^{1}H longitudinal relaxation and ^{1}H spin-lock decay



Supplementary Figure S1. (a) Bulk ^{1}H longitudinal relaxation rate measured by inversion-recovery. The blue line represents the fit with a mono-exponential recovery curve (T_1 = 32 ± 2.2 s). **(b)** Decay curve recorded during a ^{1}H spin-lock at 13.4 kHz RF. The blue line represents the fit with a mono-exponential decay curve ($T_{1\rho}$ = 13.9 ± 0.4 ms).

Pulse sequence to measure ^{1}H spin-lock efficiency and ^{1}H–^{13}C decoupling efficiency

Used for the measurement of ^{1}H spin-lock efficiency (Figure 1a, pulse scheme in Figure 2a) and ^{1}H–^{13}C decoupling efficiency (Figure 1b, pulse scheme in Figure 2b).

Options:

Spin-lock

l2=0	No spin-lock pulse
l2=1	^{1}H spin-lock pulse before CP

spinlock.pp

```
;spinlock.pp
; Measurement of 1H-13C decoupling efficiency and 1H spin-lock efficiency.

;Options
; l2=0 :       No spin-lock pulse
; l2=1 :       1H spin-lock pulse before CP

;--- Reference ---
;Recovery of bulk proton magnetization and sensitivity enhancement in ultra-fast Magic-Angle
;Spinning solid-state NMR, Jean-Philippe Demers, Vinesh Vijayan, Adam Lange, J Phys Chem B, 2015

;$OWNER=nmrsu

;"sp0=pl6"    ;Syntax for TopSpin 2.x            ; 1H CP power level
"spw0=plw6"   ;Syntax for TopSpin 3.x            ; 1H CP power level

;=============
; Parameters
;=============

  ;p2     1H 90° pulse length
  ;pl2    1H 90° pulse power level
  ;p16    1H spin-lock pulse length
  ;pl16   1H spin-lock pulse power level
  ;pl5   13C CP power level
  ;pl6    1H CP power level
  ;p15    CP contact time
  ;sp0    1H CP shape
  ;sp15  13C CP shape
  ;pl12   1H decoupling power level during acquisition
  ;d1     Recycling delay between experiments

;=============
; Protections
;=============

  1m
  if "p2  >   1000"  goto Problem
  if "p15 > 10m"     goto Problem
  if "aq  > 31m"     goto Problem
  if "d1  < 0.025s"  goto Problem
  goto PassParams
Problem, 1m
  print "Parameters not accepted, ending."
  goto HaltAcqu
PassParams, 1m

;===============
; Pulse program
;===============

1 ze                                          ;accumulate into an empty memory
2 d1 do:f2                                    ;recycling delay, decoupler off
  1u fq=0:f2

  1u fq=cnst15:f1                 ;frequency offset on 13C channel for band-selective CP schemes
  10u pl2:f2 pl5:f1                           ;set initial power levels

;--- 90° 1H ---
  p2:f2 ph1

;--- 1H Spin-lock ---
  if (l2==1)
  {
    1u pl16:f2                                ;1H spin-lock power level
    (p16 ph0):f2
  }

;--- 1H --> 13C Cross-polarization ---
  1u pl6:f2                                   ;1H CP power level
  (p15 ph2):f1 (p15:sp0 ph0):f2               ;shaped pulse on 1H, rectangular on 13C

  1u pl12:f2                                  ;1H decoupling power level
  1u cpds2:f2                                 ;Acquisition 1H decoupling on

;--- Acquisition ---
  go=2 ph31
```

```
  1m do:f2                                          ;decoupler off
  100m wr #0                                        ;save data to disk

HaltAcqu, 1m                                        ;jump address for protection files
  exit                                              ;quit

;=============
; Phase cycle
;=============

  ph1=  1 3                     ; 1H 90° pulse
  ph0=  0                       ; 1H CP pulse
  ph2=  0 0 2 2                 ;13C CP pulse
  ph31= 0 2 2 0                 ;13C acquisition
```

Pulse sequence for flip-back experiment

Used for the measurement of sensitivity curve (Figure 4b) and signal enhancement (Figure 5), pulse scheme in Figure 2c.

Parameters:

The following composite pulse decoupling (CPD) programs must be employed:

cpdprg1='xix.cpd', cpdprg2='cw.cpd', cpdprg3='cw0_flip270.cpd', cpdprg4='cw0_flip90.cpd'

Options:

Cross-polarization

l3=1	Rectangular pulse on both ^{1}H and ^{13}C channels
l3=2	Shaped pulse on ^{1}H, rectangular pulse on ^{13}C
l3=3	Rectangular pulse on ^{1}H, shaped pulse on ^{13}C

Decoupling

l4=0	Reference (No flip pulse)
l4=2	Flip-back pulse
l4=4	Flip-down pulse
l4=1	Use alternative CPD program as specified by cpdprg1 (e.g. 'xix.cpd')

flipback.pp

```
;flipback.pp
;  Measurement of sensitivity curve and signal enhancement.

;Options
; l3=1 :       Rectangular pulse on both 1H and 13C channels
; l3=2 :       Shaped pulse on 1H, rectangular pulse on 13C
; l3=3 :       Rectangular pulse on 1H, shaped pulse on 13C

; l4=0 :       Reference (No flip pulse)
; l4=2 :       Flip-back pulse
; l4=4 :       Flip-down pulse
; l4=1 :       Use alternative CPD program as specified by cpdprg1

;--- Reference ---
;Recovery of bulk proton magnetization and sensitivity enhancement in ultra-fast Magic-Angle
;Spinning solid-state NMR, Jean-Philippe Demers, Vinesh Vijayan, Adam Lange, J Phys Chem B, 2015

;$OWNER=nmrsu

;"sp0=pl6"    ;Syntax for TopSpin 2.x            ; 1H CP power level
"spw0=plw6"   ;Syntax for TopSpin 3.x            ; 1H CP power level
;"sp15=pl5"   ;Syntax for TopSpin 2.x            ;13C CP power level
"spw15=plw5"  ;Syntax for TopSpin 3.x            ;13C CP power level
"cnst25=0"                                       ;Counter for current scan number
```

```
;=============
; Parameters
;=============

  ;p2     1H 90° pulse length
  ;p12    1H 90° pulse power level
  ;pl5   13C CP power level
  ;pl6    1H CP power level
  ;p15    CP contact time
  ;sp0    1H CP shape
  ;sp15  13C CP shape
  ;p25    Duration of 1H CW decoupling pulse during acquisition
  ;pl25   1H CW decoupling power
  ;pl12   1H decoupling power level for alternative CPD program (cpdprg1, set l4=1)
  ;d1     Recycling delay between experiments

;=============
; Protections
;=============

  1m
  if "p2  >   1000"  goto Problem
  if "p15 > 10m"     goto Problem
  if "aq  > 31m"     goto Problem
  if "d1  < 0.025s"  goto Problem
  goto PassParams
Problem, 1m
  print "Parameters not accepted, ending."
  goto HaltAcqu
PassParams, 1m

;===============
; Pulse program
;===============

1 ze                                          ;accumulate into an empty memory
2 d1 do:f2                                    ;recycling delay, decoupler off
  1u fq=0:f2

  1u fq=cnst15:f1                 ;frequency offset on 13C channel for band-selective CP schemes
  10u pl2:f2 pl5:f1                           ;set initial power levels

;--- 90° 1H ---
  p2:f2 ph1

;--- 1H --> 13C Cross-polarization ---
  1u pl6:f2                                   ;1H CP power level
  if (l3==1)
  {
    (p15 ph2):f1 (p15 ph0):f2                 ;rectangular pulse on both channels
  }
  if (l3==2)
  {
    (p15 ph2):f1 (p15:sp0 ph0):f2             ;shaped pulse on 1H, rectangular on 13C
  }
  if (l3==3)
  {
    (p15:sp15 ph2):f1 (p15 ph0):f2            ;rectangular pulse on 1H, shaped on 13C
  }

;--- 1H Decoupling during acquisition ---
  if (l4==0)
  {
    1u cpds2:f2                               ;cw.cpd, CW decoupling pulse along X
  }
  else
  {
    if (l4==1)
    {
      1u pl12:f2
      1u cpds1:f2                             ;Alternative CPD program
    }
    else
    {
    ;--- Decoupling programs containing flip pulses ---
    ;  Phase cycling instructions contained in CPD programs are ignored by some versions
    ;    of TopSpin. To implement phase cycling of the last 90° pulse ('flip pulse'), we use
    ;    two CPD programs (cpdprg3 and cpdprg4) which alternate every scan.
    ;  For the flip-back, the 'flip' pulse always has an opposite phase relative to ph1 (3 1),
    ;    in order to flip 1H magnetization back to the Iz axis.
```

```
    ;  For the flip-down, the 'flip' pulse always has the same phase as ph1 (1 3), further
    ;    pushing the 1H magnetization down to the -Iz axis.
      if "(l4/2+cnst25+ds)%2==1"
      {
        1u cpds3:f2                                ;cw0 flip270.cpd
        ; This CPD program contains a CW pulse along X, then a 90° pulse along -Y
        ; It is executed on odd scans for flip-back, and even scans for flip-down.
      }
      if "(l4/2+cnst25+ds)%2==0"
      {
        1u cpds4:f2                                ;cw0 flip90.cpd
        ; This CPD program contains a CW pulse along X, 90° pulse along Y
        ; It is executed on even scans for flip-back, and odd scans for flip-down.
      }
      "cnst25=cnst25+1"
    }
  }

;--- Acquisition ---
  go=2 ph31
  1u do:f2                                         ;decoupler off
  100m  wr #0                                      ;save data to disk

HaltAcqu, 1m                                       ;jump address for protection files
  exit                                             ;quit

;=============
; Phase cycle
;=============

  ph1 = 1 3                    ; 1H 90° pulse
  ph0 = 0                      ; 1H CP pulse
  ph2 = 0 0 2 2                ;13C CP pulse
  ph31= 0 2 2 0                ;13C acquisition
```

Composite pulse decoupling (CPD) programs

cw.cpd

```
;$OWNER=nmrsu

;--- 1H CW decoupling/spin-lock pulse, phase along X ---

1 p25:0 pl=pl25

;--- Do not pulse for the rest of the sequence ---
2 5m
jump to 2
```

cw0_flip270.cpd

```
;$OWNER=nmrsu

;--- 1H CW decoupling/spin-lock pulse, phase along X ---

1 p25:0 pl=pl25

;--- 1H 90° flip pulse, phase along -Y ---

  p2:270 pl=pl2

;--- Do not pulse for the rest of the sequence ---
2 5m
jump to 2
```

cw0_flip90.cpd

```
;$OWNER=nmrsu

;--- 1H CW decoupling/spin-lock pulse, phase along X ---

1 p25:0 pl=pl25

;--- 1H 90° flip pulse, phase along Y ---

  p2:90 pl=pl2

;--- Do not pulse for the rest of the sequence ---
2 5m
jump to 2
```

xix.cpd (Example of alternative CPD program)

```
  0.3u fq=cnst21
  0.5u pl=pl12
1 pcpd:0
  pcpd:180
jump to 1
```

Pulse sequence to measure remaining magnetization and magnetization recovery curves

Used for the measurement of ^{1}H remaining magnetization (Figure 3) and recovery curve (Figure 4a), pulse scheme in Figure 2d.

Parameters:

The following composite pulse decoupling (CPD) programs must be employed:

cpdprg1='xix.cpd', cpdprg2='cw.cpd', cpdprg3='cw0_flip270.cpd', cpdprg4='cw0_flip90.cpd'

Options:

Cross-polarization

l3=1	Rectangular pulse on both ^{1}H and ^{13}C channels
l3=2	Shaped pulse on ^{1}H, rectangular pulse on ^{13}C
l3=3	Rectangular pulse on ^{1}H, shaped pulse on ^{13}C

Content of the 'mock' experiment

l5=1	Reference experiment (no 'mock' experiment executed)
l5=2	Only the ^{1}H 90° pulse and ^{1}H spin-lock are executed in the 'mock' experiment
l5=3	The ^{1}H 90° pulse and ^{1}H–^{13}C CP are executed in the 'mock' experiment
l5=4	The full 'mock' experiment (^{1}H 90°, ^{1}H–^{13}C CP and *cw* ^{1}H decoupling) is executed
l5=5	Full 'mock' experiment (reference, no flip pulse), magnetization recovery curve
l5=6	Full 'mock' experiment and flip-back pulse, magnetization recovery curve
l5=7	Full 'mock' experiment and flip-down pulse, magnetization recovery curve

Pre-saturation

l6=0	No pre-saturation
l6=2	Pre-saturation

recovery.pp

```
;recovery.pp
;  Measurement of 1H remaining magnetization and recovery curve.

;Options
; l3=1 :       Rectangular pulse on both 1H and 13C channels
; l3=2 :       Shaped pulse on 1H, rectangular pulse on 13C
; l3=3 :       Rectangular pulse on 1H, shaped pulse on 13C

; l5=1 :       Reference experiment (no 'mock' experiment executed)
; l5=2 :       Only the 1H 90° pulse and 1H spin-lock are executed in the 'mock' experiment
; l5=3 :       The 1H 90° pulse and 1H-13C CP are executed in the 'mock' experiment
; l5=4 :       The full 'mock' experiment (1H 90°, 1H-13C CP and CW 1H decoupling) is executed
; l5=5 :       Full 'mock' experiment (reference, no flip pulse), magnetization recovery curve
; l5=6 :       Full 'mock' experiment and flip-back pulse, magnetization recovery curve
; l5=7 :       Full 'mock' experiment and flip-down pulse, magnetization recovery curve

; l6=0 :       No pre-saturation
; l6=2 :       Pre-saturation
```

```
;--- Reference ---
;Recovery of bulk proton magnetization and sensitivity enhancement in ultra-fast Magic-Angle
;Spinning solid-state NMR, Jean-Philippe Demers, Vinesh Vijayan, Adam Lange, J Phys Chem B, 2015

;$OWNER=nmrsu

;"sp0=pl6"    ;Syntax for TopSpin 2.x            ; 1H CP power level
"spw0=plw6"   ;Syntax for TopSpin 3.x            ; 1H CP power level
;"sp15=pl5"   ;Syntax for TopSpin 2.x            ;13C CP power level
"spw15=plw5"  ;Syntax for TopSpin 3.x            ;13C CP power level

;=============
; Parameters
;=============

  ;p2     1H 90° pulse length
  ;p27    Duration of one high-power pre-saturation pulse (two pulses, X and -X, are executed).
  ;pl2    Power level for 1H 90° pulse and high-power pre-saturation pulses
  ;pl5   13C CP power
  ;pl6    1H CP power
  ;p15    CP contact time
  ;sp0    1H CP shape
  ;sp15  13C CP shape
  ;pl12   1H decoupling power during acquisition ('Read' experiment)
  ;p25    Duration of decoupling pulse during 'mock' experiment
  ;pl25   1H decoupling power during 'mock' experiment
  ;d1     Recycling delay before the 'mock' experiment
  ;d2     Recovery delay between the 'mock' and the 'read' experiment

;==============
; Protections
;==============

  1m
  if "p1  >   1000"  goto Problem
  if "p2  >   1000"  goto Problem
  if "p15 > 10m"     goto Problem
  if "p27 > 10m"     goto Problem
  if "aq  > 31m"     goto Problem
  if "d1  < 0.025s"  goto Problem
  goto PassParams
Problem, 1m
  print "Parameters not accepted, ending."
  goto HaltAcqu
PassParams, 1m

;===============
; Pulse program
;===============

1 ze                                             ;accumulate into an empty memory
2 1u do:f2                                       ;decoupler off
  1u fq=0:f2
  1u fq=cnst15:f1                ;frequency offset on 13C channel for band-selective CP schemes

;--- 1H Pre-saturation ---
  if (l6==2)
  {
    1u pl2:f2
    p2:f2 ph21                                   ;90° 1H pulse along Y
    p27*0.5:f2 ph10                              ;1H spin-lock presat pulse along X
    p27*0.5:f2 ph12                              ;1H spin-lock presat pulse along -X
  }
  d1                                             ;recycling delay before the 'mock' experiment

;--------------------
; 'Mock' experiment
;--------------------

;--- Mock experiment: 90° 1H ---
  if (l5<5)
  {
    1u pl2:f2
    p2:f2 ph1    ;For measurement of remaining 1H magnetization amounts, the transverse 1H
                 ;  magnetization generated by the 'mock' experiment is preserved by the phase
                 ;  cycling (ph1).
  } else {
    1u pl2:f2
    p2:f2 ph21   ;For measurement of the 1H recovery behavior, the transverse 1H magnetization
                 ;  generated by the 'mock' experiment is discarded by the phase cycling (ph21).
```

```
  }

  if (l5==2)
  {
  ;--- Mock experiment: 1H Spin-lock ---
    1u pl6:f2                                   ;1H CP power level
    if (l3==2)
    {
      (p15:sp0 ph20):f2                         ;shaped pulse on 1H
    }
    else
    {
     (p15 ph20):f2                              ;rectangular pulse on 1H
    }
  }
  if (l5>2)
  {
  ;--- Mock experiment: 1H --> 13C Cross-polarization ---
    1u pl5:f1 pl6:f2                            ;13C and 1H CP power level
    if (l3==1)
    {
      (p15 ph22):f1 (p15 ph20):f2               ;rectangular pulse on both channels
    }
    if (l3==2)
    {
      (p15 ph22):f1 (p15:sp0 ph20):f2           ;shaped pulse on 1H, rectangular on 13C
    }
    if (l3==3)
    {
      (p15:sp15 ph22):f1 (p15 ph20):f2          ;rectangular pulse on 1H, shaped on 13C
    }
  }

;--- Mock experiment: 1H decoupling ---
  if (l5>3)
  {
    1u pl25:f2                                  ;1H decoupling power for 'mock' experiment
    (p25 ph20):f2                               ;CW decoupling on 1H
  }

;--- Mock experiment: Flip pulses ---
  1u pl2:f2
  if (l5==6)
  {
    p2:f2 ph23                          ;1H 90° flip-back (opposite phase to initial 90° pulse)
  }
  if (l5==7)
  {
    p2:f2 ph21                          ;1H 90° flip-down (same phase to initial 90° pulse)
  }

;--------------------
; 'Read' experiment
;--------------------
;Detection experiment with the same CP conditions
;  as the 'mock' experiment but with XiX decoupling during acquisition

  ;--- Read experiment: Recovery delay and 90° 1H ---
  ;For measurement of the 1H recovery behavior, a variable recovery delay (d2) is inserted after
  ;  the 'mock' experiment. The 'read' experiment starts with a new 90° pulse which excites
  ;  longitudinal 1H magnetization (Iz).
  if (l5>4)
  {
    d2    ;Delay between the 'mock' and the 'read' experiment.
    p2:f2 ph1
  }

  ;For measurement of remaining 1H magnetization amounts, the 'read' experiment starts directly
  ;  after the 'mock' experiment, without any delay. The remaining transverse 1H magnetization
  ;  is then directly spin-locked by the 1H CP pulse of the 'read' experiment.

  ;--- Read experiment: 1H --> 13C Cross-polarization ---
  1u pl5:f1 pl6:f2
  if (l3==1)
  {
    (p15 ph2):f1 (p15 ph0):f2                   ;rectangular pulse on both channels
  }
  if (l3==2)
  {
    (p15 ph2):f1 (p15:sp0 ph0):f2               ;shaped pulse on 1H, rectangular on 13C
```

```
  }
  if (l3==3)
  {
    (p15:sp15 ph2):f1 (p15 ph0):f2                ;rectangular pulse on 1H, shaped on 13C
  }

  1u pl12:f2                                      ;1H decoupling power level
  1u cpds1:f2                                     ;Acquisition 1H decoupling on

;--- Acquisition ---
  go=2 ph31
  1m do:f2                                        ;decoupler off
  wr #0                                           ;save data to disk

HaltAcqu, 1m                                      ;jump address for protection files
  exit                                            ;quit

;=============
; Phase cycle
;=============

;--- 'Mock' experiment phases ---
  ph21= 1                       ;1H 90° (pre-sat. and 'mock' experiment)
  ph10= 0                       ;1H presat X
  ph12= 2                       ;1H presat X
  ph20= 0                       ;1H spin-lock
  ph22= 1                       ;13C spin-lock (SOCP)
  ph23= 3                       ;1H flip-back

;--- 'Read' experiment phases ---
  ph1 = 1 3                     ;1H 90°
  ph0 = 0 0 0 0  2 2 2 2        ;1H CP pulse
  ph2 = 0 0 2 2  0 0 2 2
        1 1 3 3  1 1 3 3        ;13C CP pulse
  ph31= 0 2 2 0  2 0 0 2
        1 3 3 1  3 1 1 3        ;13C acquisition
```

Full references

(7) Knight, M. J.; Webber, A. L.; Pell, A. J.; Guerry, P.; Barbet-Massin, E.; Bertini, I.; Felli, I. C.; Gonnelli, L.; Pierattelli, R.; Emsley, L.; Lesage, A.; Herrmann, T.; Pintacuda, G., Fast Resonance Assignment and Fold Determination of Human Superoxide Dismutase by High-Resolution Proton-Detected Solid-State MAS NMR Spectroscopy. *Angew. Chem. Int. Edit.* **2011,** *50*, 11697-11701.

(19) Wickramasinghe, N. P.; Parthasarathy, S.; Jones, C. R.; Bhardwaj, C.; Long, F.; Kotecha, M.; Mehboob, S.; Fung, L. W. M.; Past, J.; Samoson, A.; Ishii, Y., Nanomole-Scale Protein Solid-State NMR by Breaking Intrinsic ^{1}H T_1 Boundaries. *Nat. Methods* **2009,** *6*, 215-218.

(24) Marchetti, A.; Jehle, S.; Felletti, M.; Knight, M. J.; Wang, Y.; Xu, Z.-Q.; Park, A. Y.; Otting, G.; Lesage, A.; Emsley, L.; Dixon, N. E.; Pintacuda, G., Backbone Assignment of Fully Protonated Solid Proteins by ^{1}H Detection and Ultrafast Magic-Angle-Spinning NMR Spectroscopy. *Angew. Chem. Int. Edit.* **2012,** *51*, 10756-10759.

(25) Barbet-Massin, E.; Pell, A. J.; Jaudzems, K.; Franks, W. T.; Retel, J. S.; Kotelovica, S.; Akopjana, I.; Tars, K.; Emsley, L.; Oschkinat, H.; Lesage, A.; Pintacuda, G., Out-and-Back ^{13}C–^{13}C Scalar Transfers in Protein Resonance Assignment by Proton-Detected Solid-State NMR under Ultra-Fast MAS. *J. Biomol. NMR* **2013,** *56*, 379-386.

(32) Barbet-Massin, E.; Pell, A. J.; Retel, J. S.; Andreas, L. B.; Jaudzems, K.; Franks, W. T.; Nieuwkoop, A. J.; Hiller, M.; Higman, V.; Guerry, P.; Bertarello, A.; Knight, M. J.; Felletti, M.; Le Marchand, T.; Kotelovica, S.; Akopjana, I.; Tars, K.; Stoppini, M.; Bellotti, V.; Bolognesi, M.; Ricagno, S.; Chou, J. J.; Griffin, R. G.; Oschkinat, H.; Lesage, A.; Emsley, L.; Herrmann, T.; Pintacuda, G., Rapid Proton-Detected NMR Assignment for Proteins with Fast Magic Angle Spinning. *J. Am. Chem. Soc.* **2014,** *136*, 12489-12497.

(33) Barnes, A. B.; De Paepe, G.; van der Wel, P. C. A.; Hu, K. N.; Joo, C. G.; Bajaj, V. S.; Mak-Jurkauskas, M. L.; Sirigiri, J. R.; Herzfeld, J.; Temkin, R. J.; Griffin, R. G., High-Field Dynamic Nuclear Polarization for Solid and Solution Biological NMR. *Appl. Magn. Reson.* **2008,** *34*, 237-263.

(34) Maly, T.; Debelouchina, G. T.; Bajaj, V. S.; Hu, K. N.; Joo, C. G.; Mak-Jurkauskas, M. L.; Sirigiri, J. R.; van der Wel, P. C. A.; Herzfeld, J.; Temkin, R. J.; Griffin, R. G., Dynamic Nuclear Polarization at High Magnetic Fields. *J. Chem. Phys.* **2008,** *128*, 052211.